


RCL-vaihtovirtapiiri: resonanssi

Olkoon tarkastelun kohteena tavallinen RLC-vaihtovirtapiiri. Piirissä on kolme komponenttia, R ohmin vastus, L henryn induktanssi ja C faradin kapasitanssi.



Piiriin syötettyyn jännitteeseen kohdistuu kolme eri pudotusta,

$$\begin{aligned} E_L &= L \left(\frac{d}{dt} I(t) \right) && \text{käämin yli,} \\ E_R &= R I(t) && \text{vastuksen yli, sekä} \\ E_C &= \frac{1}{C} \int I(t) dt && \text{kondensaattorin yli.} \end{aligned}$$

Näiden summasta syntyy piirin Kirchhoffin lain mukainen sähkömotorinen voima $E(t)$. Mikäli piiriin syötetty jännite on sinimuotoista vaihtovirtaa $E(t) = E_0 \sin(\omega t)$, saadaan piirille yhtälö

$$E(t) = L \left(\frac{d}{dt} I(t) \right) + R I(t) + \frac{1}{C} \int I(t) dt = E_0 \sin(\omega t).$$

Integraalitermistä päästään eroon derivoimalla yhtälö ajan suhteen:

$$L \left(\frac{d^2}{dt^2} I(t) \right) + R \left(\frac{d}{dt} I(t) \right) + \frac{I(t)}{C} = E_0 \omega \cos(\omega t)$$

Tämä on differentiaaliyhtälö pakotetulle värähtelylle, jossa ulkoinen pakottava jännite on sinimuotoinen amplitudina E_0 .

Jos $E(t)=0$, kyseessä vapaa värähtelypiiri ja differentiaaliyhtälö on homogeeninen.

Jos piirissä ei ole vastusta, ts. $R = 0$, piiri on vaimentamaton. Vastuksen olemassaolo merkitsee, että piirissä on vaimennus.

Tarkastellaan differentiaaliyhtälön ratkaisuja vakiodien L , R , C , E_0 ja ω eri arvoilla.

Aluksi hävitetään mahdollisista aiemmista laskuista jääneet muuttujat:

[> **restart;**

[Virtaa kuvaava toisen kertaluvun yhtälö.

[> **yhtalo:= L*diff(i(t), t\$2)+R*diff(i(t), t)+1/C*i(t)=omega*E[0]*cos(omega*t);**

$$yhtalo := L \left(\frac{d^2}{dt^2} i(t) \right) + R \left(\frac{d}{dt} i(t) \right) + \frac{i(t)}{C} = \omega E_0 \cos(\omega t)$$

Vaimentamaton tapaus

[Vaimentamattomassa tapauksessa piirissä ei ole vastusta ja yhtälö on

[> **yhtalo1:= subs(R=0, yhtalo);**

$$yhtalo1 := L \left(\frac{d^2}{dt^2} i(t) \right) + \frac{i(t)}{C} = \omega E_0 \cos(\omega t)$$

[Alkuehtona olkoon, että piirissä ei tapahdu mitään:

[> **alkuehto:= i(0)=0, D(i)(0)=0;**

$$alkuehto := i(0) = 0, D(i)(0) = 0$$

[Tämän ratkaisu on

[> **dsolve({yhtalo1, alkuehto}, i(t)):**
ratkaisu1:= subs(%, i(t));

$$ratkaisu1 := \frac{\cos\left(\frac{t}{\sqrt{C} \sqrt{L}}\right) C E_0 \omega}{-1 + \omega^2 C L} - \frac{\omega E_0 \cos(\omega t) C}{-1 + \omega^2 C L}$$

[Ratkaisu muodostuu kahdesta kosinitermistä: $\cos\left(\frac{t}{\sqrt{L} C}\right)$ on peräisin homogeeniyhtälön yleisestä ratkaisusta ja $\cos(\omega t)$ epähomogeeniyhtälön yksittäisratkaisusta. Edellinen kuvaa piirin sisäistä värähtelyä, jälkimmäinen ulkoisen jännitteen taajuudella tapahtuvaa värähtelyä.

[Ratkaisu tässä muodossa ei kuitenkaan ole pätevä, jos nimittäjä tulee nolllaksi.

[> **assume(C>0, L>0);**

[Olettamalla kapasitanssi ja induktanssi positiivisiksi helpotetaan symbolista ratkaisua.

[> **restaajuu:= solve(-1+omega^2*L*C=0, omega);**

$$restaajuu := \frac{1}{\sqrt{C \sim L \sim}}, -\frac{1}{\sqrt{C \sim L \sim}}$$

Nollaksi tuloa vastaavaa pakotteen taajuutta kutsutaan *resonanssitaajuudeksi*. Tällöin piirin sisäinen värähtelytaajuus ja pakotteen taajuus ovat samat. Kyseessä on tilanne, missä epähomogeenisen yhtälön yksittäisratkaisu saakin erilaisen muodon:

```
> yhtalo2:= subs(omega=sqrt(C*L)/(C*L), yhtalo1);
```

$$yhtalo2 := L \left(\frac{d^2}{dt^2} i(t) \right) + \frac{i(t)}{C} = \frac{\sqrt{C L} E_0 \cos\left(\frac{\sqrt{C L} t}{C L}\right)}{C L}$$

```
> dsolve({yhtalo2, alkuehto}, i(t)):
ratkaisu2:= subs(%,i(t)):
simplify(%);
```


$$\frac{1}{2} \frac{E_0 \sin\left(\frac{t}{\sqrt{C} \sqrt{L}}\right) t}{L}$$

Tämä on tulkittavissa värähtelytermiksi, jossa virran amplitudi $E_0 t / (2 L)$ kasvaa rajatta muuttujan t mukana. Käytännössä tämä johtaisi piirin palamiseen.

Arvoilla $L = 0.1 \text{ H}$, $C = 0.001 \text{ }\mu\text{F}$, $E_0 = 230 \text{ V}$ saadaan resonanssitaajuudella seuraava kuvio:

```
> subs({L=0.1, C=0.001, E[0]=230}, ratkaisu2);
plot(%, t = 0..1.5);
```

$$1150.000000 \sin(99.99999999 t) t$$




Resonanssitaajuus on

```
> subs({L=0.1, C=0.001}, restaajuus[1]);  
100.
```

Jos pakotteen taajuus on lähellä resonanssitaajuutta, saadaan erikoinen värähtely:

```
> subs({L=0.1, C=0.001, E[0]=230, omega=90}, ratkaisu1);  
plot(%, t = 0..1.5);  
-108.9473684 cos(99.99999999 t) + 108.9473684 cos(90 t)
```



Tässä pakote aluksi vahvistaa piirissä kulkevaa virtaa kuten resonanssitaaajuuden tapauksessakin, mutta koska taajuudet eivät olekaan täsmälleen samat, värähtelyt siirtyvät vähitellen vastakkaisiin vaiheisiin, ja pakote alkaa sammuttaa piirin virtaa.

Syntynyt värähtely voidaan myös ymmärtää sinivärähtelyksi, jolla on sinimuotoinen vaihteleva amplitudi:

```
> (cos(100*t)-cos(90*t))-(-2*sin(5*t)*sin(95*t)):
simplify(%);
```

0

Vaimennettu tapaus

Yleisen vaimennetun tapauksen ratkaisu on

```
> yhtalo;
```

$$L \left(\frac{d^2}{dt^2} i(t) \right) + R \left(\frac{d}{dt} i(t) \right) + \frac{i(t)}{C} = \omega E_0 \cos(\omega t)$$

```
> dsolve(yhtalo, i(t)):
subs(%, i(t)):
ratkaisu:= collect(combine(%), {_C1, _C2, exp});
```

$$\text{ratkaisu} := e^{\left(-\frac{(C \sim R - \sqrt{C^2 R^2 - 4 C \sim L \sim}) t}{2 C \sim L \sim} \right)} _C2 + e^{\left(-\frac{(C \sim R + \sqrt{C^2 R^2 - 4 C \sim L \sim}) t}{2 C \sim L \sim} \right)} _C1$$

$$+ \frac{\omega E_0 \cos(\omega t) C \sim - C \sim^2 \omega^3 E_0 \cos(\omega t) L \sim + C \sim^2 \omega^2 E_0 \sin(\omega t) R}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2}$$

Ratkaisu koostuu kahdesta termistä:

```
> sisainen:= select(has, ratkaisu, _C1)+select(has, ratkaisu,
_C2);
```

$$\text{sisainen} := e^{\left(-\frac{(C \sim R + \sqrt{C \sim^2 R^2 - 4 C \sim L \sim}) t}{2 C \sim L \sim} \right)} _C1 + e^{\left(-\frac{(C \sim R - \sqrt{C \sim^2 R^2 - 4 C \sim L \sim}) t}{2 C \sim L \sim} \right)} _C2$$

```
> ulkoinen:= subs({_C1=0, _C2=0}, ratkaisu);
```

$$\text{ulkoinen} := \frac{\omega E_0 \cos(\omega t) C \sim - C \sim^2 \omega^3 E_0 \cos(\omega t) L \sim + C \sim^2 \omega^2 E_0 \sin(\omega t) R}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2}$$

Sisäisen värähtelyn termi esittää vaimenevaa värähtelyä, sillä muuttujan t kerroin kummassakin eksponentissa on joko negatiivinen reaaliiluku tai kompleksiluku, jonka reaaliiosa on negatiivinen. Nämä termit siis kuolevat vähitellen pois.

Ulkoisen värähtelyn termi voidaan muokata seuraavasti. Pyritään saattamaan se muotoon, jossa on vain yksi sinifunktio kertoimen sopiva *amplitudi* A ja argumentissa sopiva *vaihesiirto* δ .

```
> uusimuoto:= expand(A*sin(omega*t+delta));
```

$$\text{uusimuoto} := A \sin(\omega t) \cos(\delta) + A \cos(\omega t) \sin(\delta)$$

Vaativalla että sini- ja kosinitermien kertoimet vanhassa ja uudessa esitysmuodossa ovat samat, saadaan ehdot, joista pyritään ratkaisemaan A ja δ .

```
> ehdot:= zip((x, y)->x=y, map2(coeff, ulkoinen, [sin(omega*t),
cos(omega*t)]), map2(coeff, uusimuoto, [sin(omega*t),
cos(omega*t)])):
simplify(%)
```

$$\left[\begin{aligned} \frac{C \sim^2 \omega^2 E_0 R}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} &= A \cos(\delta), \\ -\frac{C \sim \omega E_0 (-1 + \omega^2 C \sim L \sim)}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} &= A \sin(\delta) \end{aligned} \right]$$

```
> solve({op(ehdot)}, {A, delta}):
ratk:= allvalues(%)
```

$$\text{ratk} := \left\{ A = \sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2}} E_0 \omega C \sim, \delta = \arctan \left(\right. \right.$$

$$\begin{aligned}
& - \frac{-1 + \omega^2 C \sim L \sim}{\sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} (R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2)}}, \\
& \left. \frac{C \sim \omega R}{\sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} (R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2)}} \right\} \\
& , \left\{ \delta = \arctan \left(\right. \right. \\
& \left. \frac{-1 + \omega^2 C \sim L \sim}{\sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} (R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2)}}, \right. \\
& \left. - \frac{C \sim \omega R}{\sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} (R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2)}} \right) \\
& , A = - \sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2}} E_0 \omega C \sim \left. \right\}
\end{aligned}$$

Pakotteen aiheuttama virta on siis sinimuotoinen amplitudina

> amplitudi := subs(ratk[1], A);

$$amplitudi := \sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2}} E_0 \omega C \sim$$

Amplitudi riippuu pakotteen taajuudesta. Pyritään määrittämään maksimiampplitudi ja vastaava taajuus:

> diff(amplitudi, omega):
derivaatta := simplify(%);

derivaatta :=

$$\begin{aligned}
& \frac{(\omega^4 C \sim^2 L \sim^2 - 1) E_0 C \sim}{\sqrt{\frac{1}{R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2} (R^2 C \sim^2 \omega^2 + 1 - 2 \omega^2 C \sim L \sim + \omega^4 C \sim^2 L \sim^2)}^2}
\end{aligned}$$

> maksimikohta := solve(derivaatta=0, omega);

$$maksimikohta := \frac{I}{\sqrt{C \sim L \sim}}, \frac{-I}{\sqrt{C \sim L \sim}}, \frac{1}{\sqrt{C \sim L \sim}}, -\frac{1}{\sqrt{C \sim L \sim}}$$

> subs(omega=sqrt(C*L)/(C*L), amplitudi);

```
maksimiampplitudi:= simplify(%);
```

$$\frac{\sqrt{\frac{L}{R^2 C}} E_0 \sqrt{C L}}{L}$$

$$maksimiampplitudi := \frac{\overline{\text{csgn}(R)} E_0}{R}$$

Maple on varovainen yhdistellessään neliöjuuria, ja ainakaan ohjelman versio 6 ei suostunut sieventämään lauseketta pidemmälle. Lauseke on kuitenkin yksinkertainen ja käsin sieventämällä päästään tulokseen $\frac{E_0}{R}$. Maksimiarvo saadaan siis resonanssitaajuudella.

Piirretään pakotteen (ulkoisen jännitteen) ja resonanssitaajuutta vastaavan virran kuvaajat:

```
> numulkoinen:= subs({L=0.1, C=0.0001, R=5, E[0]=230}, ulkoinen);
```

$$numulkoinen := \frac{0.0230 \omega \cos(\omega t) - 0.230 \cdot 10^{-6} \omega^3 \cos(\omega t) + 0.00001150 \omega^2 \sin(\omega t)}{-0.00001975 \omega^2 + 1. + 0.1 \cdot 10^{-9} \omega^4}$$


```
> jannite:= subs(E[0]=230, E[0]*sin(omega*t));
```

$$jannite := 230 \sin(\omega t)$$

```
> res:= subs({L=0.1, C=0.0001, R=5, E[0]=230}, 1/sqrt(L*C));
```

$$res := 316.2277660$$


```
> plot(subs(omega=res, {jannite, numulkoinen}), t=0..0.1);
```



Kuviossa on jännite punaisella ja virta vihreällä.

Vastaava kuvio, kun kyseessä ei ole resonanssitaajuus:

```
> plot(subs(omega=280, {jannite, numulkoinen}), t=0..0.1);
```



```
>
```

Tehtäviä

Tutki, miten alkuehtojen muuttaminen vaikuttaa ratkaisuun vaimentamattomassa tapauksessa. Miten ratkaisun luonne muuttuu, kun ollaan resonanssikohdan lähellä?

Piirrä kuvaaja, joka esittää vaimennetun tapauksen virtaa pakotteen taajuuden funktiona. Onko vaimennetun piirin tapauksessa sellainen ratkaisu mahdollinen, jossa virran amplitudi rajatta kasvaa, ts. piiri palaa?

Linkit

[vaihtovirtapiirin pakotettu värähtely](#) (virtap3.mws)
[epähomogeeninen vakiokertoiminen lineaariyhtälö](#)

JP & SKK & MS 12.07.2001